

```
/*
 *      A finished version of the pong game
 *      Made by Ole Marius Hoel Rindal
 *      Updated 03.05.12
 */

//Variables
ballx = 0;
bally = 0;
player1x = 0;
player1y = 0;
player2x = 0;
player2y = 0;
player1Score = 0;
player2Score = 0;
batSpeed = 5;
ballSpeedX = 5;
ballSpeedY = 2;

//Keys to control Player 1
//Arrow up and down
player1Up = 38;
player1Down = 40;

//Keys to control Player 2
//Arrow up and arrow down on number pad
//Remember to turn off Num Lock!!
player2Up = 224;
player2Down = 225;

ballImg = loadImage("ball.jpg");
batImg = loadImage("bat.png");

ball = toSprite(ballImg);
player1 = toSprite(batImg);
player2 = toSprite(batImg);

//This part runs your program
startRound();
while TRUE do
    player1Move();
    player2Move();
    moveBall();
    wait(20);
done;

//Starts a new round
function startRound()

    //Place the ball at the center of screen
    ballx = width() / 2;
    bally = height() / 2;
    spriteVisible(ball,TRUE);
    spriteMove(ball,ballx,bally);

    //Place the left bat at center on left side
    player1x = width() - 635;
    player1y = height() / 2;
    spriteVisible(player1,TRUE);
```

```
spriteMove(player1,player1x,player1y);

//Place the right bat at center on right side
player2x = width() - 15;
player2y = height() / 2;
spriteVisible(player2,TRUE);
spriteMove(player2,player2x,player2y);

//Print the score
println(player1Score," - ",player2Score);
endfunc;

//Move the bat for Player 1
function player1Move()

//Check if up or down key is pressed and check if bat is outside screen
if keyPressed(player1Up) and player1y < height()-height(player1) then
    player1y = player1y + batSpeed;
elseif keyPressed(player1Down) and player1y > 0 then
    player1y = player1y - batSpeed;
endif;

spriteMove(player1,player1x,player1y);
endfunc;

//Move the bat for Player 2
function player2Move()

//Check if up or down key is pressed and check if bat is outside screen
if keyPressed(player2Up) and player2y < height()-height(player2) then
    player2y = player2y + batSpeed;
elseif keyPressed(player2Down) and player2y > 0 then
    player2y = player2y - batSpeed;
endif;

spriteMove(player2,player2x,player2y);
endfunc;

//A player scores a goal
//pl - indicates which player who scored
function goal(pl)

//Check who scored and print it out
if pl == 1 then
    player1Score = player1Score + 1;
    cls();
    println("Player1 score!");
    wait(1000);
else
    player2Score = player2Score + 1;
    println("Player2 score!");
    wait(1000);
endif;

cls();           //Clears the screen
startRound();    //Restart round
endfunc;

//Move the ball
function moveBall()
    //Move the ball
```

```
ballx = ballx + ballSpeedX;
bally = bally + ballSpeedY;
spriteMove(ball,ballx,bally);

//If ball hits top or bottom
if bally == height() then
    ballSpeedY = -ballSpeedY;
elseif bally == 0 then
    ballSpeedY = -ballSpeedY;
endif;

//If ball hits bats
if spriteCollision(player1,ball) or spriteCollision(player2,ball) then
    ballSpeedX = -ballSpeedX;
endif;

//If ball hits left end of screen player one scores a goal
if ballx == width()-width(ball) then
    goal(1)
endif;

//If ball hits right end of screen player two scores a goal
if ballx == 0 then
    goal(2)
endif
endfunc;
```
